
Szanowni Państwo ! 
 
Oddajemy do rąk Państwa kolejną wersję naszego dokumentu opisującego zagadnienia związane z 
korzystaniem z oprogramowania SAS w polskich warunkach czyli aspekty lokalizacyjne SAS-a. 
Niniejsza jego wersja jest wprawdzie kontynuacją poprzednich dokumentów, przygotowanych dla 
wersji 8, ale zawartością, a również tytułem znacznie różni się od poprzednich – podobnie jak różni się 
SAS 9.1 od SAS-ów 8.  
Występujące w tytule określenie „SAS Foundation” oznacza te elementy SAS-a 9, które zostały 
odziedziczone po SAS-ie 8 i rozwinięte drogą ewolucyjną.  
Migracja SAS 8 do SAS 9 Foundation (tzw. „1 w 1”) jest względnie łatwa i nie powinna powodować 
większych problemów również w dziedzinie NLS-u (National Language Support), do czego ten 
dokument chce się przyczynić.  
Dodatkiem do opisu zagadnień lokalizacyjnych SAS-a Foundation 9.1.3 jest rozdział o SPDE – nowym 
motorze dostępnym w ramach SAS/BASE. 
SAS-a 9.1.3 w systemie Windows można zainstalować i uruchamiać w polskiej wersji językowej. Taka 
instalacja spowoduje, że SAS naturalnie dostosowuje się do specyfiki naszego języka i kraju. 
Niniejszy dokument wskazuje na napotkane przez nas na etapie testów odstępstwa od tej zasady, 
wymagające takiej czy innej uwagi. W wybranych sytuacjach umieściliśmy opis poprawnego działania 
elementów oprogramowania, aby zachęcić Państwa do samodzielnego sprawdzenia. W polskim 
biurze SAS Institute testowaliśmy i omawiamy w dokumencie polską wersję instalacji SAS-a 9.1 na 
Windowsach.  
Aspekty lokalizacyjne instalacji na UNIXach są omawiane w osobnym dokumencie o nazwie 
„Polonizacja SAS-a Foundation 9.1.3. Dodatek dla platformy UNIX”. 
„Polonizacja SAS-a Foundation 9.1.3 platformie Windows” powinna zostać udostępniona każdej 
osobie stykającej się z zagadnieniami związanymi z polskimi znakami w tworzonych aplikacjach. 
Lektura może oszczędzić wysiłek w czasie tworzenia i uruchamiania aplikacji. 
 
Jednocześnie zaznaczamy, że dokument ten jest uzupełnieniem, a nie zastąpieniem dokumentacji. W 
szczególności polecamy publikację SAS National Language Support (NLS): User’s Guide dostępną 
dla wszystkich naszych Klientów w SAS-owym helpie lub OnlineDocu. 
 
Dokument i inne informacje związane z tematem będą publikowane na naszych stronach WWW pod 
adresem: 

http://www.sas.com/poland/services 
 

W razie dodatkowych pytań prosimy o kontakt z naszym Zespołem, telefonicznie na numer 
(22) 5604666 lub elektronicznie pod adresem support@spl.sas.com.  
 
Prosimy o uwagi i spostrzeżenia, które pomogą nam w przygotowaniu następnych wersji tego 
dokumentu.  
 

Zawsze służąc Państwu pomocą, 
Zespół Pomocy Technicznej Centrum Technologii  SAS Polska 

http://www.sas.com/poland/services
mailto:support@spl.sas.com

1. BASE 
 
1.0.  Polska instalacja SAS-a 
 
Podczas instalacji wybieramy jakie wersje narodowe chcemy zainstalować. Podstawowym 
wymaganiem instalacji polskiej wersji SAS-a są polskie ustawienia regionalne w Windowsach. 
Proponujemy oprócz zainstalowania wersji polskiej zainstalować również angielską. 
Polska wersja SAS-a ma polskie menu, polskie okienka oraz w znacznym zakresie polskie 
komunikaty.  
 
Uruchomienie polskiej wersji SAS-a, np. ikoną „SAS 9.1 (Polish)” nie jest konieczne, aby korzystać z 
udogodnień przy przetwarzaniu z językiem polskim (polskie znaki, daty, itd.). Niezbędne jest tylko 
ustawienie opcji locale.  
Na polskich Windowsach (istotne są polskie ustawienia regionalne) opcja locale jest podczas instalacji 
automatycznie ustawiona na "Polish" (co jest równoważne ustawieniu pl_Pl  lub Polish_Poland) 
również dla angielskiej wersji SAS-a 9.1 (uruchamianej ikoną „SAS 9.1 (English)”). Ustawienie tej opcji 
znajduje się w sasowym configu (dla polskiej instalacji w <folder instalacyjny SASa>\nls\pl\sasv9.cfg). 
 
Opcja locale odzwierciedla lokalne konwencje języka, kraju lub regionu. Ustawia polskie czcionki, 
formę przedstawiania daty, czasu, liczb i waluty. 
 
Wraz z ustawienim opcji locale automatycznie ustawiane są opcje: encoding, trantab, dflang, 
datestyle, papersize, co ułatwia użytkownikom pracę z językiem narodowym. 
 
UWAGA: SAS (r) 9.1 (TS1M3) 
UWAGA: Sesja działa na platformie XP_PRO . 
 
1    proc options option=locale; 
LOCALE=POLISH_POLAND 
                   Określa bieżące ustawienia lokalne sesji SAS-a. 
 
2    proc options option=encoding; 
ENCODING=WLATIN2  Określa domyślne kodowanie dla przetwarzania danych zewnętrznych. 
 
3    proc options option=trantab; 
TRANTAB=(lat1lat1,lat1lat1,wlt2_ucs,wlt2_lcs,wlt2_ccl,,,) 
                   Nazwy tablic translacji 
 
4    proc options option=dflang; 
DFLANG=POLISH     Język europejskich formatów i informatów daty i godziny 
 
5    proc options option=datestyle; 
DATESTYLE=YMD     Identyfikuje sekwencje miesiąca, dnia i roku z niejednoznacznym informatem 
                   ANYDATE 
6    proc options option=papersize; 
7    run; 
PAPERSIZE=LETTER  Format papieru do drukowania 
 
Pomimo ustawienia opcji papersize na letter do drukowania okienek sasowych jest wybierany 
poprawny format A4. 
Opcja encoding jest jedyną z powyższej listy, której wartość nie może być zmieniona w trakcie sesji. 
 
Oprócz opcji locale szczególną rolę pełni opcja encoding. Ustawienie tej opcji decyduje o interpretacji 
kodów znaków w czasie przetwarzania.  Każdy zbiór w formacje SAS 9 posiada w nagłówku 
informacje o stronie kodowej oraz o platformie systemu operacyjnego na jakiej został zapisany.  
 
Oto przykład nagłówka zbioru: 
                                       Procedura CONTENTS 
 
    Nazwa zbioru            WORK.AIR                             Obserwacje             144 
    Typ elementu            DATA                                 Zmienne                2 
    Mechanizm               V9                                   Indeksy                0 
    Utworzono               środa, 23 czerwiec 2004 15:44:03     Długość obserwacji     16 
    Ostatnio zmodyfikowano  środa, 23 czerwiec 2004 15:44:03     Usuniętych obserwacji  0 
    Zabezpieczenie                                               Skompresowany          Nie 
    Typ zbioru                                                   Posortowany            Nie 

    Etykieta 
    Reprezentacja danych    WINDOWS_32 
    Kodowanie               wlatin2  Central Europe (Windows) 
 
 
                     Informacje zależne od mechanizmu/systemu operacyjnego 
 
              Rozmiar strony zbioru              4096 
              Liczba stron zbioru                1 
              Pierwsza strona                    1 
              Maksymalnie obserwacji na stronie  252 
              Obserwacji na pierwszej stronie    144 
              Liczba napraw zbioru               0 
              Nazwa pliku                        C:\work\sas\_TD1596\air.sas7bdat 
              Utworzone przez wersję             9.0101M3 
              System operacyjny                  XP_PRO 
 
 
                            Alfabetyczna lista zmiennych i atrybutów 
 
       #    Zmienna    Typ     Dł.    Format    Etykieta 
 
       2    AIR        Num       8              international airline travel (thousands) 
       1    DATE       Num       8    MONYY. 
 
Bieżące ustawienie opcji encoding (na poziomie sesji, libname, filename, pojedynczego zbioru) nie 
tylko decyduje o stronie kodowej nowo utworzonego zbioru czy pliku zewnętrznego, ale także może 
decydować o interpretacji odczytywanych danych i ewentualnym przekodowaniu czytanych zbiorów 
(jeżeli np. były tworzone na innej platformie).  
Automatyczne przekodowania danych między różnymi encodingami (transcoding) wykonuje 
mechanizm CEDA (Cross Environment Data Exchange). Aby w logu otrzymywać informacje o 
ewentualnych przekodowaniach należy ustawić opcję: 
 
options msglevel=i; 
 
Dodatkowo można też ustawić opcję biblioteki inencoding i outencoding wymuszając encoding 
zbiorów czytanych i zapisywanych do niej zbiorów. 
W przypadku częstej konieczności pracy z różnymi encodingami zachęcamy do lektury dokumentacji 
SAS-a, w szczególności SAS Products, SAS National Language Support: User's Guide, rozdziały: 
NLS Concepts, Encoding for NLS. 
 
Encoding można też ustawić tak, aby obsługiwał jeden ze standardów Unicode. Uwagi na temat 
używania Unicodu znajdują się w rozdziale 1.8. 
 
Występujący w wersjach 8 SAS-a problem niemożności uruchomienia przez tego samego 
użytkownika dwóch SAS-ów  z ustawioną opcją locale został rozwiązany i nie ma konieczności 
stosowania obejścia (ustawienie opcji rsasuser). 
 
 
1.1. DMS – Display Manager System 
 
Nie powinno brakować polskich czcionek ani w menu sasowym „po polsku”, ani w okienkach 
sasowych. Gdyby u Państwa ich nie było należy sprawdzić, czy wybrana czcionka ma ustawiony 
skrypt ‘Środkowoeuropejski’ lub ‘Europa Środkowa’. Czcionkę w Enhanced Edytorze ustawiamy w 
Narzędzia-> Opcje-> Enhanced Editor-> Wygląd. Czcionkę w okienkach Edytor programów, Log i 
Wynik wybieramy przez Narzędzia-> Opcje-> Czcionka przy zaznaczonym jednym z tych okienek.  
 
Jeżeli zmienimy czcionkę to możemy takie ustawienia zapamiętać dla następnych sesji SAS-a, jeżeli 
zaznaczymy pole „Zapisuj ustawienia końcowe” w Narzędzia-> Opcje-> Preferencje. 
 
 
1.1.1. Explorer (SAS-owy) 
 
Sortowanie Windows-owych folderów lub plików (w Favorite Folders i przez View-> Details) nie działa 
po polsku. 

1.1.2. Viewtable 
 
Nie działają zapytania (Dane-> Warunek…) do zbioru otwartego w viewtable. Przy konstrukcji 
zapytania nie pojawiają się nazwy kolumn, za to pojawiają się za to błędy w logu. Problem występuje 
tylko w polskiej wersji SAS-a. Problem został rozwiązany w Service Packu 2 do SAS-a. Posiadamy też 
odpowiedniego fixa, którego można ściągnąć ze strony polskiego biura. 
 
 
1.2 Polskie sortowanie. 
 
Aby sortować po polsku wystarczy ustawić opcję ‘sortseq’. 
Najprościej jest ustawić ją domyślnie dla wszystkich użytkowników w sasv9.cfg (<folder instalacyjny 
SASa>\nls\pl\sasv9.cfg): 
 
-sortseq Polish 
 
Dla testu można wykonać: 
 
* jeżeli opcja sortseq jest ustawiona globalnie to poniższa linia jest zbędna; 
options sortseq=Polish; 
 
data t; 
  length dane $20; 
  input dane @@; 
  cards; 
zbir żuk Monika ą źdźbło maszyna Lew łopata Ą Ala analiza lewica Łódź Adam 
; 
run; 
proc sort data=t; 
by dane; 
title1 'Zbiór po posortowaniu standardowym - komputerowym'; 
proc print; 
run; 
 
Procedura SORT będzie, po ustawieniu opcji sortseq, wykonywała polskie sortowanie komputerowe 
AĄBCĆ...aąbcć (analogiczne do porządku w znakach ASCII). W przypadku potrzeby sortowania w 
innym porządku np. AaĄąBbCcĆć... udostępniamy makro, które stworzy potrzebną tablice sortowania.  
SAS sortuje komputerowo, porównując kolejne znaki. Jest to trochę inny porządek niż w słowniku. 
Przykładem niech będzie posortowany przez SAS-a ciąg: Kot kat kot. W sortowaniu językowym (np. w 
słowniku) oczekiwalibyśmy kat kot Kot. Implementacja takiego sortowania jest planowana w 
przyszłych wersjach. 
 
Rozważamy (Centrum Technologii w biurze polskim), czy sortowania AaĄąBbCcĆć nie uczynić 
domyślnym (zamiast AĄBCĆ...aąbcć). Państwa uwagi będą dla nas cenną wskazówką. 
 
 
1.3 Polskie formaty i funkcje: liczbowe, daty, godziny, waluty. 
 
Ustawienie opcji locale na Polish (domyślne dla polskiej instalacji SASa) pozwala na używanie 
polskich formatów. 
 
Przykłady przedstawienia liczby z przecinkiem dziesiętnym: 
 
data _null_; 
l=-3384.70; 
put l; 
put l numx12.2; *bez spacji między tysiącami; 
put l nlnum12.2; *ze spacją między tysiącami - przy locale=Polish; 
put l commax12.2;  
*długość formatów jest przykładowa ; 
run; 

Przykłady użycia polskich formatów daty: 
 
data _null_; 
data='7oct1960'd; 
put data; 
put data DATE.; put 'i po polsku:'; 
put data POLDFDE.;  
put data POLDFDT.;  
put data POLDFDD.;  
put data POLDFDWN.; 
put data POLDFMN.; 
put data POLDFMY.; 
put data POLDFWKX.;   
put data POLDFDN.; 
put data POLDFWDX.; 
amount=1; 
put 'przeliczenie 1 euro na pln - biżący kurs można skonfigurować:' amount 
eurtoplz10.2 ' zł';  
amount=10; 
put amount eurfrplz9.2; 
run; 
 
Inną możliwością uzyskania tych samych efektów jest używanie odpowiednich formatów EURxxxx. 
Ponieważ w polskiej instalacji SAS-a jest ustawiona opcja dflang na Polish, formaty EURxxxx. będą 
interpretowane jako polskie. 
 
Nową w SAS 9.1 grupą formatów i informatów są te zaczynające się od NLxxxx podobnie jak 
EURxxxx zależne od ustawienia opcji dflang (lub locale). Przykłady: 
 
data _null_; 
data='7oct1960'd; 
put data NLDATE.; * DATE or WORDDATE equivalent; 
put data NLDATEMN.; * MONNAME equivalent; 
put data NLDATEW.; * WEEKDATE equivalent; 
put data NLDATEWN.; * Print day of week name; 
put data NLDATM.; * DATETIME equivalent; 
put data NLDATMTM.; * TOD equivalent; 
put data NLDATMW.; * TWMDY equivalent; 
amount=123456; 
put amount nlmny15.2; 
amount=123456; 
put amount euro15.2;  
put amount nlmnieur15.2; * alternatywny sposób pokazania symbolu euro;  
run; 
 
Oprócz formatów można też używać szeregu informatów i funkcji uwzględniających polskie konwencje 
językowe. 
 
W przypadku używania formatów, których domyślne długości są niewystarczające dla wyników w 
języku polskim należy explicite podać długość formatu. W przeciwnym wypadku SAS może, np. 
zamiast pełnej nazwy wypisać mnemonik. Tę sytuację ilustruje następujący fragment logu. 
 
1    options locale=Polish_Poland; 
2 
3    data _null_; 
4    i=3; 
5    weekname=put(i,nldatewn12.); 
6    put weekname; 
7    weekname=put(i,nldatewn.); 
8    put weekname; 
9    run; 
 
poniedziałek 
Pn 

O formatach i informatach związanych z Unicodem wspominamy w rozdziale 1.8. 
 
 
1.4 ODS 
 
W rezultatach zdefiniowanych przez polecenie ods (np. html, rtf, pdf) widać polskie znaki. 
W przypadku ods PDF SAS-a trzeba wcześniej przygotować. Sposób postępowania opisuje nota SAS 
Note SN-009650.  
Aby przybliżyć Państwu tę notę proponujemy przykład stworzenia szablonu, przy wykorzystaniu 
którego rezultat ods pdf będzie posiadał polskie znaki. 
 
/*  
Uruchom jednorazowo dla instalacji SAS-a zmieniając fontpath zależnie od wersji 
Windows:  
proc fontreg; fontpath 'c:\windows\fonts'; *dla Windows XP;  run;  
*/ 
 
proc template; 
define style nlsfonts; 
parent=styles.default; 
replace fonts / 
 'TitleFont2' = ("<ttf> Arial, Helvetica, sans-serif",4,Bold Italic) 
      'TitleFont' = ("<ttf> Arial, Helvetica, sans-serif",5,Bold Italic) 
      'StrongFont' = ("<ttf> Arial, Helvetica, sans-serif",4,Bold) 
      'EmphasisFont' = ("<ttf> Arial, Helvetica, sans-serif",3,Italic) 
      'FixedEmphasisFont' = ("<ttf> Courier New, Courier, monospace",2,Italic) 
      'FixedStrongFont' = ("<ttf> Courier New, Courier, monospace",2,Bold) 
      'FixedHeadingFont' = ("<ttf> Courier New, Courier, monospace",2) 
      'BatchFixedFont' = ("<ttf> SAS Monospace, Courier New, Courier, monospace",2) 
      'FixedFont' = ("<ttf> Courier",2) 
      'headingEmphasisFont' = ("<ttf> Arial, Helvetica, sans-serif",4,Bold Italic) 
      'headingFont' = ("<ttf> Arial, Helvetica, sans-serif",4,Bold) 
      'docFont' = ("<ttf> Arial, Helvetica, sans-serif",3); 
  end; 
run; 
 
.... 
ods pdf file='c:\temp\test.pdf' style=nlsfonts; 
.... 
 
Template zostanie domyślnie stworzony w bibliotece SASUSER. Odpowiedź na pytanie jak 
udostępnić go innym użytkownikom znajduje się w poniższym linku: 
http://support.sas.com/faq/033/FAQ03376.html 
 
 
1.4.1 PDF generowany z aplikacji klienckich (np. Enterprise Guide, SAS Olap Viewer) 
 
Nie każdy PDF wygenerowany przez SAS-a jest tworzony przy użyciu mechanizmu ODS. Część 
aplikacji klienckich wykorzystuje do tego celu mechanizm tzw. wirtualnych drukarek PDF. Ustawienia 
dotyczące tego mechanizmu zawarte są w rejestrze SAS-a. Niestety, domyślne ustawienia wirtualnej 
drukarki PDF nie zezwalają na utworzenie pliku PDF z poprawnie wyświetlanymi polskimi znakami. 
Potrzebna jest modyfikacja rejestru SAS-a. Poniższy kod realizuje przykładową modyfikację rejestru i 
pozwala na utworzenie poprawnych plików PDF w aplikacjach klienckich. 
 
/* Zarejestruj wszystkie czcionki systemowe w SAS-ie */ 
proc fontreg; 
 fontpath "%sysget(systemroot)\Fonts"; 
run; 
 
/* Wyeksportuj wszystkie czcionki zarejestrowane w bibliotece SASUSER do pliku 
   fonts.sasxreg */ 
proc registry export="D:\fonts.sasxreg" startat="core\printing\freetype\fonts";  
run; 

http://support.sas.com/faq/033/FAQ03376.html

/* Importuj zawartość pliku fonts.sasxreg do biblioteki SASHELP */ 
proc registry import="D:\fonts.sasxreg" usesashelp; 
run; 
 
/* W pliku ttfpdf.sasxreg utwórz aliasy dla rodziny czcionek PDF */ 
data _null_; 
 file "D:\ttfpdf.sasxreg"; 
 input line $80.; 
 put line; 
 lines; 
[CORE\PRINTING\PDF\FONT FAMILIES] 
"AvantGardeGothic"="<ttf> MS Gothic" 
"Bookman"="<ttf> Times New Roman" 
"Chancery"="<ttf> Times New Roman" 
"Courier"="<ttf> Courier New" 
"Dingbats"="<ttf> Wingdings" 
"Helvetica"="<ttf> Arial" 
"NewCenturySchoolbook"="<ttf> Times New Roman" 
"Palatino"="<ttf> Palatino Linotype" 
"Symbol"="<ttf> Symbol" 
"Times"="<ttf> Times New Roman" 
[CORE\PRINTING\ALIAS\FONTS\PDF] 
"Century Schoolbook"="<ttf> Century Schoolbook" 
"NewCenturySchlbk"="<ttf> Times New Roman" 
"Times New Roman"="<ttf> Times New Roman" 
"ZapfChancery"="<ttf> Times New Roman" 
"ZapfDingbats"="<ttf> Wingdings" 
"ITC Bookman"="<ttf> Times New Roman" 
"Courier"="<ttf> Coutier New" 
"SAS Monospace"="<ttf> Courier" 
"Palatino"="<ttf> Palatino Linotype" 
"Times"="<ttf> Times New Roman" 
"monospace"="<ttf> Courier New" 
"san-serif"="<ttf> Arial" 
"serif"="<ttf> Times New Roman" 
[CORE\PRINTING\PRINTERS\PDF\ADVANCED] 
"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 
[CORE\PRINTING\PRINTERS\PDF\DEFAULT SETTINGS] 
"Font Typeface"="<ttf> Courier New" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.3 (Partial Font Embedding)\ADVANCED] 
"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.3 (Full Font Embedding)\DEFAULT SETTINGS] 
"Font Typeface"="<ttf> Courier New" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.3 (Full Font Embedding)\ADVANCED] 
"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.3\DEFAULT SETTINGS] 
"Font Typeface"="<ttf> Courier New" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.3\ADVANCED] 
"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.2 (Partial Font Embedding)\DEFAULT 
SETTINGS] 
"Font Typeface"="<ttf> Courier New" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.2 (Partial Font Embedding)\ADVANCED] 
"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.2 (Full Font Embedding)\DEFAULT SETTINGS] 
"Font Typeface"="<ttf> Courier New" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.2 (Full Font Embedding)\ADVANCED] 
"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.2\DEFAULT SETTINGS] 
"Font Typeface"="<ttf> Courier New" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.2\ADVANCED] 
"FONT FAMILIES"=LINK:"\\CORE\\PRINTING\\PDF\\FONT FAMILIES" 
[CORE\PRINTING\PROTOTYPES\PDF Version 1.3 (Partial Font Embedding)\DEFAULT 
SETTINGS] 
"Font Typeface"="<ttf> Courier New" 
run; 
 



Polonizacja SAS-a Foundation 9.1.3 na platformie Windows 
   Str. 9 
 

SAS Institute TECHNICAL SUPPORT 
 

/* Importuj aliasy rodziny czcionek PDF do rejestru SAS-a */ 
proc registry import="D:\ttfpdf.sasxreg" usesashelp;  
run; 
 
 
 
1.5 Polskie drukowanie 
 
Na wydruku z okienek tekstowych polskie czcionki powinny się pojawiać się bez żadnych ustawień. 
Jeżeliby się nie pojawiały, należy w ustawieniach drukarki Plik-> Drukuj-> Ustawienia-> Czcionka 
zmienić czcionkę na taką, która polskie znaki posiada, pamiętając o zaznaczeniu odpowiedniego 
skryptu. Nowa czcionka zostanie zapamiętana  jako domyślna dla drukowania. Alternatywnie można 
ustawić opcję sysprintfont. 
 
 
 
1.6 Polskie nazwy zmiennych 
 
Jeżeli chcemy, aby w nazwach zmiennych  występowały polskie znaki należy ustawić opcje 
validvarname=any. Ostrzeżenie, które otrzymamy wówczas w logu, może nas powstrzymać od 
szerszego używania nazw zmiennych z polskimi znakami: 
 
1    options validvarname=any; 
OSTRZEŻENIE: Only Base procedures and SAS/STAT procedures have been tested for use with 
             VALIDVARNAME=ANY. Other use of this option is considered experimental and may 
             cause undetected errors. 
 
 
 
1.7 SPDE i „polskie ogonki” 
 
W obecnej wersji motor SPDE nie w pełni wspiera mechanizmy przekodowania pomiędzy platformami 
(transcoding) co powoduje, m.in. że nie można podłączyć przez mechanizm RLS biblioteki z sesji na 
innej platformie operacyjnej (np. z Unixa). Obejściem tego problemu może być używanie w bibliotece 
sasowej view, które odwołuje się do biblioteki spde. Ten problem nie występuje w przypadku 
korzystania z biblioteki spdsowej. 
 
 
 
1.8 Unicode 
 
Dane w zbiorze mogą być też zapisane w jednym ze standardów Unicode (utf-8). Zastosowanie 
takiego rozwiązania można sobie wyobrazić w sytuacji, gdy chcemy w zmiennych tekstowych 
zapisywać słowa z języków o różnych encodingach (np. wlatin2 i wlatin1 – odpowiednio Grażyna i 
Jürgen).  
Stosując Unicode powinniśmy pamiętać m.in. o tym, że na Unicodzie 

• operują inne funkcje tekstowe – zaczynają się od ‘k’ 
• inne formaty – zaczynają się od ‘u’  
• zmienne tekstowe mogą zajmować więcej miejsca niż zadeklarowaliśmy, gdyż 1 litera może 

być opisana nie 1B jak w przypadku  WLantinx, ale może zajmować do 4B. 
 
Aby SAS przetwarzał w standardzie Unicode należy ustawić opcję encoding=utf-8 (w configu lub przy 
wywołaniu SAS-a) i mieć zainstalowanego SAS-a (np. dodatkowo) ze wsparciem dla DBCS. Opisuje 
to nota: 
http://support.sas.com/techsup/unotes/SN/016/016223.html 
Do odczytu zbioru tekstowego w Unicode nie jest to konieczne.  
Wystarczy użyć opcji encoding w instrukcjach: filename lub (w data stepie) (in)file. Dostępne są 
wartości: utf-8, utf-16 i utf-32. 
  
Ani moduł SAS/Graph, ani SAS/AF nie wspierają standardu Unicode. 
 

http://support.sas.com/techsup/unotes/SN/016/016223.html

2. GRAPH 
 
2.1 Czcionki systemowe (HARDWARE FONTS). 
 
Nie są już potrzebne żadne dodatkowe zabiegi aby skorzystać z czcionek TrueType dostępnych w 
systemie operacyjnym Windows. Jeżeli korzystamy z polskiej wersji Windows, to bez żadnych 
problemów uzyskamy pożądane efekty po prostu używając nazw czcionek jako wartości odpowiednich 
opcji. Dodając do nazwy czcionki /bo, uzyskamy pogrubienie, a /it - czcionkę pochyłą.  
Przykładowy program demonstrujący użycie czcionek: 
 
data klasa; 
  length wzrost 8 kategoria $ 10; 
  input  wzrost kategoria $; 
  kategoria_u=upcase(kategoria);  kategoria_l=lowcase(kategoria); 
  label kategoria_u='Kat. ąęćłńóśźż'   kategoria_l='Kat. ĄĘĆŁŃÓŚŹŻ'; 
cards; 
181     Liść 
167     Sęk 
180     Źdźbło 
181     Źdźbło 
172     Kość 
181     Dąb 
; 
run; 
 
goptions reset=all; 
title1 f='Courier/it' h=10 pct 'Kat. ąęćłńóśźż'; 
title2 f='Tahoma/bo' h=6 pct 'Kat. ĄĘĆŁŃÓŚŹŻ'; 
legend1 label=(font='Arial/bo/it' 'Legenda ĄĘĆŁŃÓŚŹŻ') 
        value=(font='Arial/it/' height=1.5 pct); 
axis1   label=(font='Times New Roman/bo/it' 'Axis ĄĘĆŁŃÓŚŹŻ');      
goptions ftext='Arial'; 
proc gchart data= klasa; 
  vbar kategoria_l /type=mean sumvar=wzrost  
        maxis=axis1 
        subgroup=kategoria_u legend=legend1; 
  run; 
quit; 
 
Uzyskana na ekranie grafika zostanie poprawnie wydrukowana lub zapisana w jednym z wielu 
formatów graficznych (m.in. GIF, WMF, JPEG). Bez żadnych problemów można jej też użyć w 
aplikacjach SAS/AF i SAS/EIS.  
W identyczny sposób (zmieniając jedynie urządzenie docelowe) możemy wygenerować grafikę 
bezpośrednio na drukarkę lub do pliku zewnętrznego. 
 
 
2.2 Czcionki SAS/GRAPH (SOFTWARE FONTS). 
 
Niektóre czcionki SAS/Graph można wykorzystać do produkcji grafiki z polskimi literami niezależnie 
od systemu. Czcionki zawierające nasze „ogonki” to CENTX, SWISS, ZAPF. Te właśnie nazwy 
powinny zostać użyte w specyfikacjach czcionek. Dodatkowo należy ustawić odpowiednią tablicę 
przekodowań (keymapę) ze strony systemowej (Windows 1250) na wewnętrzną stronę kodową 
SAS/GRAPH. Tablicę przekodowań tworzymy wykonując poniższy kod, ale najpierw musimy 
dowiązać bibliotekę o nazwie gfont0, w której zostanie wygenerowana tablica przekodowań (kod na 
następnej stronie). 
 
/* generowanie keymapy dla WINDOWS */ 
libname gfont0 (sashelp); 
 
proc gkeymap name=winansi out=temp keymap; 
run; 
data mod; 
    retain tolen 1; 



Polonizacja SAS-a Foundation 9.1.3 na platformie Windows 
   Str. 11 
 

SAS Institute TECHNICAL SUPPORT 
 

    from="d3"X;TO="08"x;output;  /* Oacute   */ 
    from="f3"X;TO="a2"x;output;  /* oacute   */ 
    from="f1"X;TO="c0"x;output;  /* nacute   */ 
    from="d1"X;TO="c1"x;output;  /* Nacute   */ 
    from="9c"X;TO="c2"x;output;  /* sacute   */ 
    from="8c"X;TO="c3"x;output;  /* Sacute   */ 
    from="9f"X;TO="c4"x;output;  /* zacute   */ 
    from="8f"X;TO="c5"x;output;  /* Zacute   */ 
    from="bf"X;TO="c6"x;output;  /* zdot     */ 
    from="af"X;TO="c7"x;output;  /* Zdot     */ 
    from="b3"X;TO="c8"x;output;  /* lslash   */ 
    from="a3"X;TO="c9"x;output;  /* Lslash   */ 
    from="b9"X;TO="ca"x;output;  /* ahook   */ 
    from="a5"X;TO="cb"x;output;  /* Ahook   */ 
    from="ea"X;TO="cc"x;output;  /* ehook   */ 
    from="ca"X;TO="cd"x;output;  /* Ehook   */ 
    from="c6"X;TO="f4"x;output;  /* Cacute  */ 
    from="e6"X;TO="f9"x;output;  /* cacute  */ 
run; 
proc sort data=mod; by from; 
run; 
data temp; 
  update temp mod; by from; 
run; 
proc gkeymap name=win1250 data=temp keymap; 
run; 
 
A oto przykładowy kod korzystający z polskich znaków: 
 
goptions reset=all; 
libname gfont0 (sashelp); 
goptions keymap=win1250; 
proc gslide; 
  note f=swiss  h=3 'Żołądek'; 
  run; 
quit; 
 
Alternatywnie w SAS-ie 9 keymapy i devmapy można stworzyć makrem %lsgraph(wlt2); 
 
 
2.3 Mapa Polski. 
 
Podstawowy zbiór z mapą - Poland przedstawia podział na powiaty. Dostępny jest również zbiór 
Poland2, który zawiera powiązania powiatów z województwami. Aby otrzymać mapę z 
województwami, należy odpowiednio te dwa zbiory połączyć. Następnie użyć procedury GREMOVE, 
aby usunąć granice powiatów. Przedstawia to poniższy kod: 
 
/* SN-011083 */ 
data poland(drop=oldlong oldlat); 
set maps.poland(rename=(long=oldlong lat=oldlat)); 
long=-oldlat; 
lat=-oldlong; 
run; 
 
data poland; 
merge poland maps.poland2; 
by id; 
run; 
proc sort data=poland; 
by wojid id; 
run; 
proc gremove data=poland out=poland;  
id id;  
by wojid;  
run; 

Aby zobaczyć nową mapę w SAS-ie można np.: 
 
data dane; 
do i=1 to 16; 
  wojid=i; value=i;  
  output; 
end; 
drop i; 
run; 
 
goptions reset=all; 
proc gmap data=dane map=poland; 
id wojid; 
choro value/levels=16; 
run; 
quit; 
 
 
2.4 Polskie sortowanie na wykresach SAS/GRAPH . 
 
W tej wersji nie ma bezpośredniej możliwości posortowania wartości na osiach po polsku. Możliwe są 
co najmniej 2 obejścia tej sytuacji: 
• Stworzenie na podstawie posortowanej kolumny (z polskimi znakami) formatu i wyświetlanie 

sformatowanej wartości za pomocy opcji descrete (W EG opcja: One bar for each unique data 
value). 

• Stworzenie makrozmiennej z posortowaną listą wartości. W instrukcji axis ustawić opcję order 
wartością makromiennej. W proc gchart odwołać się do tej osi. 

 
 
3. CONNECT i SHARE 
 
O ile tylko sesja SAS-a odbierająca dane (np. connectowy serwer w przypadku proc upload) będzie 
miała ustawioną opcję locale na Polish (opcja trantab zostanie ustawiona automatycznie), nie powinni 
Państwo napotkać na tym obszarze problemów.  
Nie ma potrzeby generowania tablic przez Locale Setup Window – jak to było w poprzednich wersjach 
SAS-a. W przypadku komunikacji SAS 9 – SAS 9 z ewentualną translacją poradzi sobie mechanizm 
CEDA. W przypadku komunikacji SAS 9 – SAS 8 odpowiednie tablice zostaną automatycznie 
udostępnione. 
 
 
4. AF i EIS 
 
4.1 Prezentacja grafiki SAS/Graph 
Obiekty SAS/Graph Output i Graph Output Control bez problemów pokazują grafikę z polskimi 
znakami wygenerowanymi przy użyciu czcionek sprzętowych i wymienionych w punkcie 2.2 czcionek 
SAS/Graph. Obiekty Graphic Text i Graphic Text Control wyświetlają listę czcionek obejmującą 
czcionki z punktu 2.2 i pozwalają na używanie polskich znaków pod warunkiem odpowiednich 
ustawień opcji KEYMAP (p. 2.2).   
 
 
5. MDDB 
 
Nie ma możliwości skorzystania przy tworzeniu MDDB z ustawionej opcji SORTSEQ. Jeżeli chcemy, 
by zmienne klasyfikujące były przechowywane posortowane w zbiorze MDDB to w deklaracji CLASS 
należy użyć opcji DSORDER, a zbiór na podstawie którego tworzymy MDDB wcześniej posortować 
(p. 1.2). 
Nową – alternatywną technologią korzystania ze struktur wielowymiarowych w SAS-ie 9 jest OLAP 
Server. Omówienie zagadnień polonizacyjnych z nim związanych jest poza zasięgiem tematycznym 
tego dokumentu. 

6. Zlokalizowane produkty. 
 
Dział NLS naszego biura zlokalizował lub lokalizuje następujące produkty: 

a) SAS 9.1.3 

b) ADS 9.1.3 

c) AMO 2.1 

d) EG 4.1 

e) FMS 4.2 

f) FXA 4.2 

g) IDP 3.1 

h) SES 1.2 

i) SMA 2.2 

j) SPM 2.2 

k) WON 1.3 

l) WOJ 3.1 

m) WRS 3.1 

 
Rozszyfrowanie skrótów: 
ADS Application Delivery Studio 
AMO Add-In for Microsoft Office 
EG Enterprise Guide 
FMS Financial Management Solutions 
FXA Financial Management Add-In for Microsoft Office 
IDP Information Delivery Portal 
SES Solution Services 
SMA Solution Services Add-In for MS Office 
SPM Strategic Perfomance Management 
WOJ Web OLAP Viewer for Java 
WON Web OLAP Viewer for .Net 
WRS Web Report Studio 


